

## Legend

Read
Write
General
Functions
Schema
Performance
Multidatabase
Security

## Syntax

Read query structure
<code>[MATCH WHERE]</code>
<code>[OPTIONAL MATCH WHERE]</code>
<code>[WITH [ORDER BY] [SKIP] [LIMIT]]</code>
<code>RETURN [ORDER BY] [SKIP] [LIMIT]</code>

MATCH <a href="#">↗</a>
<code>MATCH (n:Person)-[:KNOWS]-&gt;(m:Person)</code> <code>WHERE n.name = 'Alice'</code>
Node patterns can contain labels and properties.
<code>MATCH (n)--&gt;(m)</code>
Any pattern can be used in MATCH.
<code>MATCH (n {name: 'Alice'})--&gt;(m)</code>
Patterns with node properties.
<code>MATCH p = (n)--&gt;(m)</code>
Assign a path to p.
<code>OPTIONAL MATCH (n)-[r]-&gt;(m)</code>
Optional pattern: nulls will be used for missing parts.

WHERE <a href="#">↗</a>
<code>WHERE n.property &lt;&gt; \$value</code>
Use a predicate to filter. Note that WHERE is always part of a MATCH, OPTIONAL MATCH or WITH clause. Putting it after a different clause in a query will alter what it does.
<code>WHERE EXISTS {</code> <code>  MATCH (n)--&gt;(m) WHERE n.age = m.age</code> <code>}</code>
Use an existential subquery to filter.

Write-only query structure
<code>(CREATE   MERGE)*</code>
<code>[SET DELETE REMOVE FOREACH]*</code>
<code>[RETURN [ORDER BY] [SKIP] [LIMIT]]</code>

Read-write query structure
<code>[MATCH WHERE]</code>
<code>[OPTIONAL MATCH WHERE]</code>
<code>[WITH [ORDER BY] [SKIP] [LIMIT]]</code>
<code>(CREATE   MERGE)*</code>
<code>[SET DELETE REMOVE FOREACH]*</code>
<code>[RETURN [ORDER BY] [SKIP] [LIMIT]]</code>

CREATE <a href="#">↗</a>
<code>CREATE (n {name: \$value})</code>
Create a node with the given properties.
<code>CREATE (n \$map)</code>
Create a node with the given properties.
<code>UNWIND \$listOfMaps AS properties</code>
<code>CREATE (n) SET n = properties</code>
Create nodes with the given properties.
<code>CREATE (n)-[r:KNOWS]-&gt;(m)</code>
Create a relationship with the given type and direction; bind a variable to it.
<code>CREATE (n)-[:LOVES {since: \$value}]-&gt;(m)</code>
Create a relationship with the given type, direction, and properties.

SET <a href="#">↗</a>
<code>SET n.property1 = \$value1,</code> <code>  n.property2 = \$value2</code>
Update or create a property.
<code>SET n = \$map</code>
Set all properties. This will remove any existing properties.
<code>SET n += \$map</code>
Add and update properties, while keeping existing ones.
<code>SET n:Person</code>
Adds a label Person to a node.

Database management <a href="#">↗</a>
<code>CREATE OR REPLACE DATABASE myDatabase</code>
(★) Create a database named myDatabase. If a database with that name exists, then the existing database is deleted and a new one created.
<code>STOP DATABASE myDatabase</code>
(★) Stop the database myDatabase.
<code>START DATABASE myDatabase</code>
(★) Start the database myDatabase.
<code>SHOW DATABASES</code>
List all databases in the system and information about them.
<code>SHOW DATABASE myDatabase</code>
List information about the database myDatabase.
<code>SHOW DEFAULT DATABASE</code>
List information about the default database.
<code>DROP DATABASE myDatabase IF EXISTS</code>
(★) Delete the database myDatabase, if it exists.

RETURN <a href="#">↗</a>
<code>RETURN *</code>
Return the value of all variables.
<code>RETURN n AS columnName</code>
Use alias for result column name.
<code>RETURN DISTINCT n</code>
Return unique rows.
<code>ORDER BY n.property</code>
Sort the result.
<code>ORDER BY n.property DESC</code>
Sort the result in descending order.
<code>SKIP \$skipNumber</code>
Skip a number of results.
<code>LIMIT \$limitNumber</code>
Limit the number of results.
<code>SKIP \$skipNumber LIMIT \$limitNumber</code>
Skip results at the top and limit the number of results.
<code>RETURN count(*)</code>
The number of matching rows. See Aggregating functions for more.

WITH <a href="#">↗</a>
<code>MATCH (user)-[:FRIEND]-(friend)</code>
<code>WHERE user.name = \$name</code>
<code>WITH user, count(friend) AS friends</code>
<code>WHERE friends &gt; 10</code>
<code>RETURN user</code>
The WITH syntax is similar to RETURN. It separates query parts explicitly, allowing you to declare which variables to carry over to the next part.
<code>MATCH (user)-[:FRIEND]-(friend)</code>
<code>WITH user, count(friend) AS friends</code>
<code>ORDER BY friends DESC</code>
<code>  SKIP 1</code>
<code>  LIMIT 3</code>
<code>RETURN user</code>
ORDER BY, SKIP, and LIMIT can also be used with WITH.

UNION <a href="#">↗</a>
<code>MATCH (a)-[:KNOWS]-&gt;(b)</code>
<code>RETURN b.name</code>
<code>UNION</code>
<code>MATCH (a)-[:LOVES]-&gt;(b)</code>
<code>RETURN b.name</code>
Returns the distinct union of all query results. Result column types and names have to match.
<code>MATCH (a)-[:KNOWS]-&gt;(b)</code>
<code>RETURN b.name</code>
<code>UNION ALL</code>
<code>MATCH (a)-[:LOVES]-&gt;(b)</code>
<code>RETURN b.name</code>
Returns the union of all query results, including duplicated rows.

MERGE <a href="#">↗</a>
<code>MERGE (n:Person {name: \$value})</code>
<code>  ON CREATE SET n.created = timestamp()</code>
<code>  ON MATCH SET</code>
<code>    n.counter = coalesce(n.counter, 0) + 1,</code>
<code>    n.accessTime = timestamp()</code>
Match a pattern or create it if it does not exist. Use ON CREATE and ON MATCH for conditional updates.
<code>MATCH (a:Person {name: \$value1}),</code>
<code>  (b:Person {name: \$value2})</code>
<code>MERGE (a)-[r:LOVES]-&gt;(b)</code>
MERGE finds or creates a relationship between the nodes.
<code>MATCH (a:Person {name: \$value1})</code>
<code>MERGE</code>
<code>  (a)-[r:KNOWS]-&gt;(b:Person {name: \$value3})</code>
MERGE finds or creates subgraphs attached to the node.

REMOVE <a href="#">↗</a>
<code>REMOVE n:Person</code>
Remove a label from n.
<code>REMOVE n.property</code>
Remove a property.

User management <a href="#">↗</a>
<code>CREATE USER alice SET PASSWORD \$password</code>
Create a new user and a password. This password must be changed on the first login.
<code>ALTER USER alice SET PASSWORD \$password CHANGE NOT REQUIRED</code>
Set a new password for a user. This user will not be required to change this password on the next login.
<code>ALTER USER alice SET STATUS SUSPENDED</code>
(★) Change the user status to suspended. Use SET STATUS ACTIVE to reactivate the user.
<code>ALTER CURRENT USER SET PASSWORD FROM \$old TO \$new</code>
Change the password of the logged-in user. The user will not be required to change this password on the next login.
<code>SHOW USERS</code>
List all users in the system, their status, roles and if they need to change their password.
(★) Status and roles are Enterprise Edition only.
<code>DROP USER alice</code>
Delete the user.

DELETE <a href="#">↗</a>
<code>DELETE n, r</code>
Delete a node and a relationship.
<code>DETACH DELETE n</code>
Delete a node and all relationships connected to it.
<code>MATCH (n)</code>
<code>DETACH DELETE n</code>
Delete all nodes and relationships from the database.

FOREACH <a href="#">↗</a>
<code>FOREACH (r IN relationships(path)  </code> <code>  SET r.marked = true)</code>
Execute a mutating operation for each relationship in a path.
<code>FOREACH (value IN coll  </code> <code>  CREATE (:Person {name: value}))</code>
Execute a mutating operation for each element in a list.

CALL subquery <a href="#">↗</a>
<code>CALL {</code>
<code>  MATCH (p:Person)-[:FRIEND_OF]-&gt;(other:Person) RETURN p,</code>
<code>  other</code>
<code>  UNION</code>
<code>  MATCH (p:Child)-[:CHILD_OF]-&gt;(other:Parent) RETURN p,</code>
<code>  other</code>
<code>}</code>
This calls a subquery with two union parts. The result of the subquery can afterwards be post-processed.

CALL procedure <a href="#">↗</a>
<code>CALL db.labels() YIELD label</code>
This shows a standalone call to the built-in procedure db.labels to list all labels used in the database. Note that required procedure arguments are given explicitly in brackets after the procedure name.
<code>CALL java.stored.procedureWithArgs</code>
Standalone calls may omit YIELD and also provide arguments implicitly via statement parameters, e.g. a standalone call requiring one argument input may be run by passing the parameter map {input: 'foo'}.
<code>CALL db.labels() YIELD label</code>
<code>RETURN count(label) AS count</code>
Calls the built-in procedure db.labels inside a larger query to count all labels used in the database. Calls inside a larger query always requires passing arguments and naming results explicitly with YIELD.

Import <a href="#">↗</a>
<code>LOAD CSV FROM</code>
<code>'https://neo4j.com/docs/cypher-refcard/4.1/csv/artists-refcard/4.1/csv/artists.csv' AS line</code>
<code>CREATE (:Artist {name: line[1], year: toInteger(line[2])})</code>
Load data from a CSV file and create nodes.
<code>LOAD CSV WITH HEADERS FROM</code>
<code>'https://neo4j.com/docs/cypher-refcard/4.1/csv/artists-with-headers.csv' AS line</code>
<code>CREATE (:Artist {name: line.Name, year: toInteger(line.Year)})</code>
Load CSV data which has headers.

<code>USING PERIODIC COMMIT 500</code>
<code>LOAD CSV WITH HEADERS FROM</code>
<code>'https://neo4j.com/docs/cypher-refcard/4.1/csv/artists-with-headers.csv' AS line</code>
<code>CREATE (:Artist {name: line.Name, year: toInteger(line.Year)})</code>
Commit the current transaction after every 500 rows when importing large amounts of data.
<code>LOAD CSV FROM</code>
<code>'https://neo4j.com/docs/cypher-refcard/4.1/csv/artists-fieldterminator.csv'</code>
<code>AS line FIELDTERMINATOR ';' </code>
<code>CREATE (:Artist {name: line[1], year: toInteger(line[2])})</code>
Use a different field terminator, not the default which is a comma (with no whitespace around it).

<code>LOAD CSV FROM</code>
<code>'https://neo4j.com/docs/cypher-refcard/4.1/csv/artists.csv' AS line</code>
<code>RETURN DISTINCT file()</code>
Returns the absolute path of the file that LOAD CSV is processing, returns null if called outside of LOAD CSV context.
<code>LOAD CSV FROM</code>
<code>'https://neo4j.com/docs/cypher-refcard/4.1/csv/artists.csv' AS line</code>
<code>RETURN lineNumber()</code>
Returns the line number that LOAD CSV is currently processing, returns null if called outside of LOAD CSV context.

Performance <a href="#">↗</a>
<ul style="list-style-type: none"> <li>Use parameters instead of literals when possible. This allows Cypher to re-use your queries instead of having to parse and build new execution plans.</li> <li>Always set an upper limit for your variable length patterns. It's possible to have a query go wild and touch all nodes in a graph by mistake.</li> <li>Return only the data you need. Avoid returning whole nodes and relationships — instead, pick the data you need and return only that.</li> <li>Use PROFILE / EXPLAIN to analyze the performance of your queries. See <a href="#">Query Tuning</a> for more information on these and other topics, such as planner hints.</li> </ul>

(★) Role management <a href="#">↗</a>
<code>CREATE ROLE my_role</code>
Create a role.
<code>CREATE ROLE my_second_role IF NOT EXISTS AS COPY OF my_role</code>
Create a role named my_second_role, unless it already exists, as a copy of the existing my_role.
<code>GRANT ROLE my_role, my_second_role TO alice</code>
Assign roles to a user.
<code>REVOKE ROLE my_second_role FROM alice</code>
Remove a specified role from a user.
<code>SHOW ROLES</code>
List all roles in the system.
<code>SHOW POPULATED ROLES WITH USERS</code>
List all roles that are assigned to at least one user in the system, and the users assigned to those roles.
<code>DROP ROLE my_role</code>
Delete a role.

Operators <a href="#">↗</a>	
General	DISTINCT, ,, []
Mathematical	+, -, *, /, %, ^
Comparison	=, <>, <, >, <=, >=, IS NULL, IS NOT NULL
Boolean	AND, OR, XOR, NOT
String	+
List	+, IN, [x], [x .. y]
Regular Expression	=~
String matching	STARTS WITH, ENDS WITH, CONTAINS

null <a href="#">↗</a>
<ul style="list-style-type: none"> <li><code>null</code> is used to represent missing/undefined values.</li> <li><code>null</code> is not equal to <code>null</code>. Not knowing two values does not imply that they are the same value. So the expression <code>null = null</code> yields <code>null</code> and not <code>true</code>. To check if an expression is <code>null</code>, use <code>IS NULL</code>.</li> <li>Arithmetic expressions, comparisons and function calls (except <code>coalesce</code>) will return <code>null</code> if any argument is <code>null</code>.</li> <li>An attempt to access a missing element in a list or a property that doesn't exist yields <code>null</code>.</li> <li>In <code>OPTIONAL MATCH</code> clauses, <code>nulls</code> will be used for missing parts of the pattern.</li> </ul>

Predicates <a href="#">↗</a>
<code>n.property &lt;&gt; \$value</code> Use comparison operators.
<code>exists(n.property)</code> Use functions.
<code>n.number &gt;= 1 AND n.number &lt;= 10</code> Use boolean operators to combine predicates.
<code>1 &lt;= n.number &lt;= 10</code> Use chained operators to combine predicates.
<code>n:Person</code> Check for node labels.
<code>variable IS NULL</code> Check if something is <code>null</code> .
<code>NOT exists(n.property) OR n.property = \$value</code> Either the property does not exist or the predicate is <code>true</code> .
<code>n.property = \$value</code> Non-existing property returns <code>null</code> , which is not equal to anything.
<code>n["property"] = \$value</code> Properties may also be accessed using a dynamically computed property name.
<code>n.property STARTS WITH 'Tim' OR n.property ENDS WITH 'n' OR n.property CONTAINS 'goodie'</code> String matching.
<code>n.property =~ 'Tim.*'</code> String regular expression matching.
<code>(n)-[:KNOWS]-&gt;(m)</code> Ensure the pattern has at least one match.
<code>NOT (n)-[:KNOWS]-&gt;(m)</code> Exclude matches to <code>(n)-[:KNOWS]-&gt;(m)</code> from the result.
<code>n.property IN [\$value1, \$value2]</code> Check if an element exists in a list.

CASE <a href="#">↗</a>
<code>CASE n.eyes WHEN 'blue' THEN 1 WHEN 'brown' THEN 2 ELSE 3 END</code> Return <code>THEN</code> value from the matching <code>WHEN</code> value. The <code>ELSE</code> value is optional, and substituted for <code>null</code> if missing.
<code>CASE WHEN n.eyes = 'blue' THEN 1 WHEN n.age &lt; 40 THEN 2 ELSE 3 END</code> Return <code>THEN</code> value from the first <code>WHEN</code> predicate evaluating to <code>true</code> . Predicates are evaluated in order.

(★) SHOW PRIVILEGES <a href="#">↗</a>
<code>SHOW PRIVILEGES</code> List all privileges in the system, and the roles that they are assigned to.
<code>SHOW ROLE my_role PRIVILEGES</code> List all privileges assigned to a role.
<code>SHOW USER alice PRIVILEGES</code> List all privileges of a user, and the role that they are assigned to.
<code>SHOW USER PRIVILEGES</code> Lists all privileges of the currently logged in user, and the role that they are assigned to.

Patterns <a href="#">↗</a>
<code>(n:Person)</code> Node with <code>Person</code> label.
<code>(n:Person:Swedish)</code> Node with both <code>Person</code> and <code>Swedish</code> labels.
<code>(n:Person {name: \$value})</code> Node with the declared properties.
<code>()-[r {name: \$value}]-()</code> Matches relationships with the declared properties.
<code>(n)--&gt;(m)</code> Relationship from <code>n</code> to <code>m</code> .
<code>(n)--(m)</code> Relationship in any direction between <code>n</code> and <code>m</code> .
<code>(n:Person)--&gt;(m)</code> Node <code>n</code> labeled <code>Person</code> with relationship to <code>m</code> .
<code>(m)-[:KNOWS]-(n)</code> Relationship of type <code>KNOWS</code> from <code>n</code> to <code>m</code> .
<code>(n)-[:KNOWS]:LOVES-&gt;(m)</code> Relationship of type <code>KNOWS</code> or of type <code>LOVES</code> from <code>n</code> to <code>m</code> .
<code>(n)-[r]-&gt;(m)</code> Bind the relationship to variable <code>r</code> .
<code>(n)-[*1..5]-&gt;(m)</code> Variable length path of between 1 and 5 relationships from <code>n</code> to <code>m</code> .
<code>(n)-[*]-&gt;(m)</code> Variable length path of any number of relationships from <code>n</code> to <code>m</code> . (See Performance section.)
<code>(n)-[:KNOWS]-&gt;(m {property: \$value})</code> A relationship of type <code>KNOWS</code> from a node <code>n</code> to a node <code>m</code> with the declared property.
<code>shortestPath((n1:Person)-[*..6]-(n2:Person))</code> Find a single shortest path.
<code>allShortestPaths((n1:Person)-[*..6]-&gt;(n2:Person))</code> Find all shortest paths.
<code>size((n)--&gt;()-&gt;())</code> Count the paths matching the pattern.

Labels
<code>CREATE (n:Person {name: \$value})</code> Create a node with label and property.
<code>MERGE (n:Person {name: \$value})</code> Matches or creates unique node(s) with the label and property.
<code>SET n:Spouse:Parent:Employee</code> Add label(s) to a node.
<code>MATCH (n:Person)</code> Matches nodes labeled <code>Person</code> .
<code>MATCH (n:Person) WHERE n.name = \$value</code> Matches nodes labeled <code>Person</code> with the given name.
<code>WHERE (n:Person)</code> Checks the existence of the label on the node.
<code>labels(n)</code> Labels of the node.
<code>REMOVE n:Person</code> Remove the label from the node.

Maps <a href="#">↗</a>
<code>{name: 'Alice', age: 38, address: {city: 'London', residential: true}}</code> Literal maps are declared in curly braces much like property maps. Lists are supported.
<code>WITH {person: {name: 'Anne', age: 25}} AS p RETURN p.person.name</code> Access the property of a nested map.
<code>MERGE (p:Person {name: \$map.name}) ON CREATE SET p = \$map</code> Maps can be passed in as parameters and used either as a map or by accessing keys.
<code>MATCH (matchedNode:Person) RETURN matchedNode</code> Nodes and relationships are returned as maps of their data.
<code>map.name, map.age, map.children[0]</code> Map entries can be accessed by their keys. Invalid keys result in an error.

(★) Graph read privileges <a href="#">↗</a>
<code>GRANT TRAVERSE ON GRAPH * NODES * TO my_role</code> Grant <code>traverse</code> privilege on all nodes and all graphs to a role.
<code>DENY READ {prop} ON GRAPH foo RELATIONSHIP Type TO my_role</code> Deny <code>read</code> privilege on a specified property, on all relationships with a specified type in a specified graph, to a role.
<code>GRANT MATCH {*} ON GRAPH foo ELEMENTS Label TO my_role</code> Grant <code>read</code> privilege on all properties and <code>traverse</code> privilege to a role. Here, both privileges apply to all nodes and relationships with a specified label/type in the graph.

(★) Graph write privileges <a href="#">↗</a>
<code>GRANT CREATE ON GRAPH * NODES Label TO my_role</code> Grant <code>create</code> privilege on all nodes with a specified label in all graphs to a role.
<code>DENY DELETE ON GRAPH neo4j TO my_role</code> Deny <code>delete</code> privilege on all nodes and relationships in a specified graph to a role.
<code>REVOKE SET LABEL Label ON GRAPH * FROM my_role</code> Revoke <code>set label</code> privilege for the specified label on all graphs to a role.
<code>GRANT REMOVE LABEL * ON GRAPH foo TO my_role</code> Grant <code>remove label</code> privilege for all labels on a specified graph to a role.
<code>DENY SET PROPERTY {prop} ON GRAPH foo RELATIONSHIPS Type TO my_role</code> Deny <code>set property</code> privilege on a specified property, on all relationships with a specified type in a specified graph, to a role.
<code>GRANT MERGE {*} ON GRAPH * NODES Label TO my_role</code> Grant <code>merge</code> privilege on all properties, on all nodes with a specified label in all graphs, to a role.
<code>REVOKE WRITE ON GRAPH * FROM my_role</code> Revoke <code>write</code> privilege on all graphs from a role.
<code>DENY ALL GRAPH PRIVILEGES ON GRAPH foo TO my_role</code> Deny all <code>graph</code> privileges privilege on a specified graph to a role.

Lists <a href="#">↗</a>
<code>['a', 'b', 'c'] AS list</code> Literal lists are declared in square brackets.
<code>size(\$list) AS len, \$list[0] AS value</code> Lists can be passed in as parameters.
<code>range(\$firstNum, \$lastNum, \$step) AS list</code> <code>range()</code> creates a list of numbers ( <code>step</code> is optional), other functions returning lists are: <code>labels()</code> , <code>nodes()</code> , <code>relationships()</code> .
<code>MATCH p = (a)-[:KNOWS*]-&gt;() RETURN relationships(p) AS r</code> The list of relationships comprising a variable length path can be returned using named paths and <code>relationships()</code> .
<code>RETURN matchedNode.list[0] AS value, size(matchedNode.list) AS len</code> Properties can be lists of strings, numbers or booleans.
<code>list[\$idx] AS value, list[\$startIdx..\$endIdx] AS slice</code> List elements can be accessed with <code>idx</code> subscripts in square brackets. Invalid indexes return <code>null</code> . Slices can be retrieved with intervals from <code>start_idx</code> to <code>end_idx</code> , each of which can be omitted or negative. Out of range elements are ignored.
<code>UNWIND \$names AS name MATCH (n {name: name}) RETURN avg(n.age)</code> With <code>UNWIND</code> , any list can be transformed back into individual rows. The example matches all names from a list of names.
<code>MATCH (a) RETURN [(a)--&gt;(b) WHERE b.name = 'Bob'   b.age]</code> Pattern comprehensions may be used to do a custom projection from a match directly into a list.
<code>MATCH (person) RETURN person { .name, .age}</code> Map projections may be easily constructed from nodes, relationships and other map values.

List predicates <a href="#">↗</a>
<code>all(x IN coll WHERE exists(x.property))</code> Returns <code>true</code> if the predicate is <code>true</code> for all elements in the list.
<code>any(x IN coll WHERE exists(x.property))</code> Returns <code>true</code> if the predicate is <code>true</code> for at least one element in the list.
<code>none(x IN coll WHERE exists(x.property))</code> Returns <code>true</code> if the predicate is <code>false</code> for all elements in the list.
<code>single(x IN coll WHERE exists(x.property))</code> Returns <code>true</code> if the predicate is <code>true</code> for exactly one element in the list.

List expressions <a href="#">↗</a>
<code>size(\$list)</code> Number of elements in the list.
<code>reverse(\$list)</code> Reverse the order of the elements in the list.
<code>head(\$list), last(\$list), tail(\$list)</code> <code>head()</code> returns the first, <code>last()</code> the last element of the list. <code>tail()</code> returns all but the first element. All return <code>null</code> for an empty list.
<code>[x IN list   x.prop]</code> A list of the value of the expression for each element in the original list.
<code>[x IN list WHERE x.prop &lt;&gt; \$value]</code> A filtered list of the elements where the predicate is <code>true</code> .
<code>[x IN list WHERE x.prop &lt;&gt; \$value   x.prop]</code> A list comprehension that filters a list and extracts the value of the expression for each element in that list.
<code>reduce(s = "", x IN list   s + x.prop)</code> Evaluate expression for each element in the list, accumulate the results.

(★) Database privileges <a href="#">↗</a>
<code>GRANT ACCESS ON DATABASE * TO my_role</code> Grant privilege to access and run queries against all databases to a role.
<code>GRANT START ON DATABASE * TO my_role</code> Grant privilege to start all databases to a role.
<code>GRANT STOP ON DATABASE * TO my_role</code> Grant privilege to stop all databases to a role.
<code>GRANT CREATE INDEX ON DATABASE foo TO my_role</code> Grant privilege to create indexes on a specified database to a role.
<code>GRANT DROP INDEX ON DATABASE foo TO my_role</code> Grant privilege to drop indexes on a specified database to a role.
<code>DENY INDEX MANAGEMENT ON DATABASE bar TO my_role</code> Deny privilege to create and drop indexes on a specified database to a role.
<code>GRANT CREATE CONSTRAINT ON DATABASE * TO my_role</code> Grant privilege to create constraints on all databases to a role.
<code>DENY DROP CONSTRAINT ON DATABASE * TO my_role</code> Deny privilege to drop constraints on all databases to a role.
<code>REVOKE CONSTRAINT ON DATABASE * FROM my_role</code> Revoke granted and denied privileges to create and drop constraints on all databases from a role.
<code>GRANT CREATE NEW LABELS ON DATABASE * TO my_role</code> Grant privilege to create new labels on all databases to a role.
<code>DENY CREATE NEW TYPES ON DATABASE foo TO my_role</code> Deny privilege to create new relationship types on a specified database to a role.
<code>REVOKE GRANT CREATE NEW PROPERTY NAMES ON DATABASE bar FROM my_role</code> Revoke the grant privilege to create new property names on a specified database from a role.
<code>GRANT NAME MANAGEMENT ON DEFAULT DATABASE TO my_role</code> Grant privilege to create labels, relationship types, and property names on default database to a role.
<code>GRANT ALL ON DATABASE baz TO my_role</code> Grant privilege to access, create and drop indexes and constraints, create new labels, types and property names on a specified database to a role.
<code>GRANT SHOW TRANSACTION (*) ON DATABASE foo TO my_role</code> Grant privilege to list transactions and queries from all users on a specified database to a role.
<code>DENY TERMINATE TRANSACTION (user1, user2) ON DATABASES * TO my_role</code> Deny privilege to kill transactions and queries from <code>user1</code> and <code>user2</code> on all databases to a role.
<code>REVOKE GRANT TRANSACTION MANAGEMENT ON DEFAULT DATABASE FROM my_role</code> Revoke the granted privilege to list and kill transactions and queries from all users on the default database from a role.

# Neo4j Cypher Refcard 4.1

Functions <a href="#">↗</a>
<code>coalesce(n.property, \$defaultValue)</code> The first non-null expression.
<code>timestamp()</code> Milliseconds since midnight, January 1, 1970 UTC.
<code>id(nodeOrRelationship)</code> The internal id of the relationship or node.
<code>toInteger(\$expr)</code> Converts the given input into an integer if possible; otherwise it returns <code>null</code> .
<code>toFloat(\$expr)</code> Converts the given input into a floating point number if possible; otherwise it returns <code>null</code> .
<code>toBoolean(\$expr)</code> Converts the given input into a boolean if possible; otherwise it returns <code>null</code> .
<code>keys(\$expr)</code> Returns a list of string representations for the property names of a node, relationship, or map.
<code>properties(\$expr)</code> Returns a map containing all the properties of a node or relationship.

Temporal functions <a href="#">↗</a>
<code>date("2018-04-05")</code> Returns a date parsed from a string.
<code>localtime("12:45:30.25")</code> Returns a time with no time zone.
<code>time("12:45:30.25+01:00")</code> Returns a time in a specified time zone.
<code>localdatetime("2018-04-05T12:34:00")</code> Returns a datetime with no time zone.
<code>datetime("2018-04-05T12:34:00[Europe/Berlin]")</code> Returns a datetime in the specified time zone.
<code>datetime({epochMillis: 3360000})</code> Transforms 3360000 as a UNIX Epoch time into a normal datetime.
<code>date({year: \$year, month: \$month, day: \$day})</code> All of the temporal functions can also be called with a map of named components. This example returns a date from <code>year</code> , <code>month</code> and <code>day</code> components. Each function supports a different set of possible components.
<code>datetime({date: \$date, time: \$time})</code> Temporal types can be created by combining other types. This example creates a <code>datetime</code> from a <code>date</code> and a <code>time</code> .
<code>date({date: \$datetime, day: 5})</code> Temporal types can be created by selecting from more complex types, as well as overriding individual components. This example creates a <code>date</code> by selecting from a <code>datetime</code> , as well as overriding the <code>day</code> component.
<code>WITH date("2018-04-05") AS d</code> <code>RETURN d.year, d.month, d.day, d.week, d.dayOfWeek</code> Accessors allow extracting components of temporal types.

Mathematical functions <a href="#">↗</a>
<code>abs(\$expr)</code> The absolute value.
<code>rand()</code> Returns a random number in the range from 0 (inclusive) to 1 (exclusive), [0,1). Returns a new value for each call. Also useful for selecting a subset or random ordering.
<code>round(\$expr)</code> Round to the nearest integer; <code>ceil()</code> and <code>floor()</code> find the next integer up or down.
<code>sqrt(\$expr)</code> The square root.
<code>sign(\$expr)</code> 0 if zero, -1 if negative, 1 if positive.
<code>sin(\$expr)</code> Trigonometric functions also include <code>cos()</code> , <code>tan()</code> , <code>cot()</code> , <code>asin()</code> , <code>acos()</code> , <code>atan()</code> , <code>atan2()</code> , and <code>haversin()</code> . All arguments for the trigonometric functions should be in radians, if not otherwise specified.
<code>degrees(\$expr)</code> , <code>radians(\$expr)</code> , <code>pi()</code> Converts radians into degrees; use <code>radians()</code> for the reverse, and <code>pi()</code> for $\pi$ .
<code>log10(\$expr)</code> , <code>log(\$expr)</code> , <code>exp(\$expr)</code> , <code>e()</code> Logarithm base 10, natural logarithm, <code>e</code> to the power of the parameter, and the value of <code>e</code> .

(★) Role management privileges <a href="#">↗</a>
<code>GRANT CREATE ROLE ON DBMS TO my_role</code> Grant the privilege to create roles to a role.
<code>GRANT DROP ROLE ON DBMS TO my_role</code> Grant the privilege to delete roles to a role.
<code>DENY ASSIGN ROLE ON DBMS TO my_role</code> Deny the privilege to assign roles to users to a role.
<code>DENY REMOVE ROLE ON DBMS TO my_role</code> Deny the privilege to remove roles from users to a role.
<code>REVOKE DENY SHOW ROLE ON DBMS FROM my_role</code> Revoke the denied privilege to show roles from a role.
<code>GRANT ROLE MANAGEMENT ON DBMS TO my_role</code> Grant all privileges to manage roles to a role.

(★) User management privileges <a href="#">↗</a>
<code>GRANT CREATE USER ON DBMS TO my_role</code> Grant the privilege to create users to a role.
<code>GRANT DROP USER ON DBMS TO my_role</code> Grant the privilege to delete users to a role.
<code>DENY ALTER USER ON DBMS TO my_role</code> Deny the privilege to alter users to a role.
<code>REVOKE SET PASSWORDS ON DBMS FROM my_role</code> Revoke the granted and denied privileges to alter users' passwords from a role.
<code>REVOKE GRANT SET USER STATUS ON DBMS FROM my_role</code> Revoke the granted privilege to alter the account status of users from a role.
<code>REVOKE DENY SHOW USER ON DBMS FROM my_role</code> Revoke the denied privilege to show users from a role.
<code>GRANT USER MANAGEMENT ON DBMS TO my_role</code> Grant all privileges to manage users to a role.

Spatial functions <a href="#">↗</a>
<code>point({x: \$x, y: \$y})</code> Returns a point in a 2D cartesian coordinate system.
<code>point({latitude: \$y, longitude: \$x})</code> Returns a point in a 2D geographic coordinate system, with coordinates specified in decimal degrees.
<code>point({x: \$x, y: \$y, z: \$z})</code> Returns a point in a 3D cartesian coordinate system.
<code>point({latitude: \$y, longitude: \$x, height: \$z})</code> Returns a point in a 3D geographic coordinate system, with latitude and longitude in decimal degrees, and height in meters.
<code>distance(point({x: \$x1, y: \$y1}), point({x: \$x2, y: \$y2}))</code> Returns a floating point number representing the linear distance between two points. The returned units will be the same as those of the point coordinates, and it will work for both 2D and 3D cartesian points.
<code>distance(point({latitude: \$y1, longitude: \$x1}), point({latitude: \$y2, longitude: \$x2}))</code> Returns the geodesic distance between two points in meters. It can be used for 3D geographic points as well.

Duration functions <a href="#">↗</a>
<code>duration("P1Y2M10DT12H45M30.25S")</code> Returns a duration of 1 year, 2 months, 10 days, 12 hours, 45 minutes and 30.25 seconds.
<code>duration.between(\$date1,\$date2)</code> Returns a duration between two temporal instances.
<code>WITH duration("P1Y2M10DT12H45M") AS d</code> <code>RETURN d.years, d.months, d.days, d.hours, d.minutes</code> Returns 1 year, 14 months, 10 days, 12 hours and 765 minutes.
<code>WITH duration("P1Y2M10DT12H45M") AS d</code> <code>RETURN d.years, d.monthsOfYear, d.days, d.hours, d.minutesOfHour</code> Returns 1 year, 2 months, 10 days, 12 hours and 45 minutes.
<code>date("2015-01-01") + duration("P1Y1M1D")</code> Returns a date of 2016-02-02. It is also possible to subtract durations from temporal instances.
<code>duration("PT30S") * 10</code> Returns a duration of 5 minutes. It is also possible to divide a duration by a number.

String functions <a href="#">↗</a>
<code>toString(\$expression)</code> String representation of the expression.
<code>replace(\$original, \$search, \$replacement)</code> Replace all occurrences of <code>search</code> with <code>replacement</code> . All arguments must be expressions.
<code>substring(\$original, \$begin, \$subLength)</code> Get part of a string. The <code>subLength</code> argument is optional.
<code>left(\$original, \$subLength)</code> , <code>right(\$original, \$subLength)</code> The first part of a string. The last part of the string.
<code>trim(\$original)</code> , <code>lTrim(\$original)</code> , <code>rTrim(\$original)</code> Trim all whitespace, or on the left or right side.
<code>toUpperCase(\$original)</code> , <code>toLowerCase(\$original)</code> UPPERCASE and lowercase.
<code>split(\$original, \$delimiter)</code> Split a string into a list of strings.
<code>reverse(\$original)</code> Reverse a string.
<code>size(\$string)</code> Calculate the number of characters in the string.

Aggregating functions <a href="#">↗</a>
<code>count(*)</code> The number of matching rows.
<code>count(variable)</code> The number of non-null values.
<code>count(DISTINCT variable)</code> All aggregating functions also take the <code>DISTINCT</code> operator, which removes duplicates from the values.
<code>collect(n.property)</code> List from the values, ignores <code>null</code> .
<code>sum(n.property)</code> Sum numerical values. Similar functions are <code>avg()</code> , <code>min()</code> , <code>max()</code> .
<code>percentileDisc(n.property, \$percentile)</code> Discrete percentile. Continuous percentile is <code>percentileCont()</code> . The <code>percentile</code> argument is from 0.0 to 1.0.
<code>stDev(n.property)</code> Standard deviation for a sample of a population. For an entire population use <code>stDevP()</code> .

(★) Database management privileges <a href="#">↗</a>
<code>GRANT CREATE DATABASE ON DBMS TO my_role</code> Grant the privilege to create databases to a role.
<code>REVOKE DENY DROP DATABASE ON DBMS FROM my_role</code> Revoke the denied privilege to delete databases from a role.
<code>DENY DATABASE MANAGEMENT ON DBMS TO my_role</code> Deny all privileges to manage database to a role.

(★) Privilege management privileges <a href="#">↗</a>
<code>GRANT SHOW PRIVILEGE ON DBMS TO my_role</code> Grant the privilege to show privileges to a role.
<code>DENY ASSIGN PRIVILEGE ON DBMS TO my_role</code> Deny the privilege to assign privileges to roles to a role.
<code>REVOKE GRANT REMOVE PRIVILEGE ON DBMS FROM my_role</code> Revoke the granted privilege to remove privileges from roles from a role.
<code>REVOKE PRIVILEGE MANAGEMENT ON DBMS FROM my_role</code> Revoke all granted and denied privileges for manage privileges from a role.

(★) DBMS privileges <a href="#">↗</a>
<code>GRANT ALL ON DBMS TO my_role</code> Grant privilege to perform all role management, user management, database management and privilege management to a role.

Path functions <a href="#">↗</a>
<code>length(path)</code> The number of relationships in the path.
<code>nodes(path)</code> The nodes in the path as a list.
<code>relationships(path)</code> The relationships in the path as a list.
<code>[x IN nodes(path)   x.prop]</code> Extract properties from the nodes in a path.

Relationship functions <a href="#">↗</a>
<code>type(a_relationship)</code> String representation of the relationship type.
<code>startNode(a_relationship)</code> Start node of the relationship.
<code>endNode(a_relationship)</code> End node of the relationship.
<code>id(a_relationship)</code> The internal id of the relationship.

INDEX <a href="#">↗</a>
<code>CREATE INDEX FOR (p:Person) ON (p.name)</code> Create an index on the label <code>Person</code> and property <code>name</code> .
<code>CREATE INDEX index_name FOR (p:Person) ON (p.age)</code> Create an index on the label <code>Person</code> and property <code>age</code> with the name <code>index_name</code> .
<code>CREATE INDEX FOR (p:Person) ON (p.name, p.age)</code> Create a composite index on the label <code>Person</code> and the properties <code>name</code> and <code>age</code> .
<code>MATCH (n:Person) WHERE n.name = \$value</code> An index can be automatically used for the equality comparison. Note that for example <code>toLowerCase(n.name) = \$value</code> will not use an index.
<code>MATCH (n:Person) WHERE n.name IN [\$value]</code> An index can automatically be used for the <code>IN</code> list checks.
<code>MATCH (n:Person) WHERE n.name = \$value and n.age = \$value2</code> A composite index can be automatically used for equality comparison of both properties. Note that there needs to be predicates on all properties of the composite index for it to be used.
<code>MATCH (n:Person) USING INDEX n:Person(name) WHERE n.name = \$value</code> Index usage can be enforced when Cypher uses a suboptimal index, or more than one index should be used.
<code>DROP INDEX index_name</code> Drop the index named <code>index_name</code> .

CONSTRAINT <a href="#">↗</a>
<code>CREATE CONSTRAINT ON (p:Person) ASSERT p.name IS UNIQUE</code> Create a unique property constraint on the label <code>Person</code> and property <code>name</code> . If any other node with that label is updated or created with a <code>name</code> that already exists, the write operation will fail. This constraint will create an accompanying index.
<code>CREATE CONSTRAINT uniqueness ON (p:Person) ASSERT p.age IS UNIQUE</code> Create a unique property constraint on the label <code>Person</code> and property <code>age</code> with the name <code>uniqueness</code> . If any other node with that label is updated or created with a <code>age</code> that already exists, the write operation will fail. This constraint will create an accompanying index.
<code>CREATE CONSTRAINT ON (p:Person) ASSERT exists(p.name)</code> (★) Create a node property existence constraint on the label <code>Person</code> and property <code>name</code> . If a node with that label is created without a <code>name</code> , or if the <code>name</code> property is removed from an existing node with the <code>Person</code> label, the write operation will fail.
<code>CREATE CONSTRAINT node_exists ON (p:Person) ASSERT exists(p.surname)</code> (★) Create a node property existence constraint on the label <code>Person</code> and property <code>surname</code> with the name <code>node_exists</code> . If a node with that label is created without a <code>surname</code> , or if the <code>surname</code> property is removed from an existing node with the <code>Person</code> label, the write operation will fail.
<code>CREATE CONSTRAINT ON ()-[l:LIKED]-() ASSERT exists(l.when)</code> (★) Create a relationship property existence constraint on the type <code>LIKED</code> and property <code>when</code> . If a relationship with that type is created without a <code>when</code> , or if the <code>when</code> property is removed from an existing relationship with the <code>LIKED</code> type, the write operation will fail.
<code>CREATE CONSTRAINT relationship_exists ON ()-[l:LIKED]-() ASSERT exists(l.since)</code> (★) Create a relationship property existence constraint on the type <code>LIKED</code> and property <code>since</code> with the name <code>relationship_exists</code> . If a relationship with that type is created without a <code>since</code> , or if the <code>since</code> property is removed from an existing relationship with the <code>LIKED</code> type, the write operation will fail.
<code>CREATE CONSTRAINT ON (p:Person) ASSERT (p.firstname, p.surname) IS NODE KEY</code> (★) Create a node key constraint on the label <code>Person</code> and properties <code>firstname</code> and <code>surname</code> . If a node with that label is created without both <code>firstname</code> and <code>surname</code> or if the combination of the two is not unique, or if the <code>firstname</code> and/or <code>surname</code> labels on an existing node with the <code>Person</code> label is modified to violate these constraints, the write operation will fail.
<code>CREATE CONSTRAINT node_key ON (p:Person) ASSERT (p.name, p.surname) IS NODE KEY</code> (★) Create a node key constraint on the label <code>Person</code> and properties <code>name</code> and <code>surname</code> with the name <code>node_key</code> . If a node with that label is created without both <code>name</code> and <code>surname</code> or if the combination of the two is not unique, or if the <code>name</code> and/or <code>surname</code> labels on an existing node with the <code>Person</code> label is modified to violate these constraints, the write operation will fail.
<code>DROP CONSTRAINT uniqueness</code> Drop the constraint with the name <code>uniqueness</code> .